

Modularno LED rasvjetno tijelo

Antunović, Antonio

Undergraduate thesis / Završni rad

2017

Degree Grantor / Ustanova koja je dodijelila akademski / stručni stupanj: **Josip Juraj Strossmayer University of Osijek, Faculty of Electrical Engineering, Computer Science and Information Technology Osijek / Sveučilište Josipa Jurja Strossmayera u Osijeku, Fakultet elektrotehnike, računarstva i informacijskih tehnologija Osijek**

Permanent link / Trajna poveznica: <https://um.nsk.hr/um:nbn:hr:200:893730>

Rights / Prava: [In copyright](#) / [Zaštićeno autorskim pravom.](#)

Download date / Datum preuzimanja: **2024-09-19**

Repository / Repozitorij:

[Faculty of Electrical Engineering, Computer Science and Information Technology Osijek](#)



SVEUČILIŠTE JOSIPA JURJA STROSSMAYERA U OSIJEKU
ELEKTROTEHNIČKI FAKULTET
Stručni studij

MODULARNO LED RASVJETNO TIJELO
Završni rad

Antonio Antunović

Osijek, 2016.

**FERIT**FAKULTET ELEKTROTEHNIKE, RAČUNARSTVA
I INFORMACIJSKIH TEHNOLOGIJA OSIJEK**Obrazac Z1S: Obrazac za imenovanje Povjerenstva za obranu završnog rada na preddiplomskom stručnom studiju**

Osijek, 12.12.2016.

Odboru za završne i diplomske ispite**Imenovanje Povjerenstva za obranu završnog rada na preddiplomskom stručnom studiju**

Ime i prezime studenta:	Antonio Antunović
Studij, smjer:	Preddiplomski stručni studij Elektrotehnika, smjer Automatika
Mat. br. studenta, godina upisa:	A 4100, 10.10.2016.
OIB studenta:	90507341417
Mentor:	Prof.dr.sc. Željko Hocenski
Sumentor:	Doc.dr.sc. Tomislav Matić
Predsjednik Povjerenstva:	Doc.dr.sc. Ivan Aleksi
Član Povjerenstva:	Doc.dr.sc. Tomislav Matić
Naslov završnog rada:	Modularno LED rasvjetno tijelo
Znanstvena grana rada:	Procesno računarstvo (zn. polje računarstvo)
Zadatak završnog rada	Potrebno je dizajnirati modularno difuzno LED rasvjetno tijelo s mogućnošću digitalnog upravljanja intenzitetom svjetla pojedinog modula (Sumentor: dr.sc. Tomislav Matić).
Prijedlog ocjene pismenog dijela ispita (završnog rada):	Izvrstan (5)
Kratko obrazloženje ocjene prema Kriterijima za ocjenjivanje završnih i diplomskih radova:	Primjena znanja stečenih na fakultetu: 3 Postignuti rezultati u odnosu na složenost zadatka: 3 Jasnoća pismenog izražavanja: 2 Razina samostalnosti: 3
Datum prijedloga ocjene mentora:	12.12.2016.
Potpis mentora za predaju konačne verzije rada u Studentsku službu pri završetku studija:	Potpis:
	Datum:

**FERIT**FAKULTET ELEKTROTEHNIKE, RAČUNARSTVA
I INFORMACIJSKIH TEHNOLOGIJA OSIJEK**IZJAVA O ORIGINALNOSTI RADA**

Osijek, 16.12.2016.

Ime i prezime studenta:

Antonio Antunović

Studij:

Preddiplomski stručni studij Elektrotehnika, smjer Automatika

Mat. br. studenta, godina upisa:

A 4100, 10.10.2016.

Ephorus podudaranje [%]:

17

Ovom izjavom izjavljujem da je rad pod nazivom: **Modularno LED rasvjetno tijelo**

izrađen pod vodstvom mentora Prof.dr.sc. Željko Hocenski

i sumentora Doc.dr.sc. Tomislav Matić

moj vlastiti rad i prema mom najboljem znanju ne sadrži prethodno objavljene ili neobjavljene pisane materijale drugih osoba, osim onih koji su izričito priznati navođenjem literature i drugih izvora informacija. Izjavljujem da je intelektualni sadržaj navedenog rada proizvod mog vlastitog rada, osim u onom dijelu za koji mi je bila potrebna pomoć mentora, sumentora i drugih osoba, a što je izričito navedeno u radu.

Potpis studenta:


SADRŽAJ

1. UVOD.....	1
2. FUNKCIONALNI BLOK DIJAGRAM.....	2
3. KONTROLNO-UPRAVLJAČKA JEDINICA (MASTER)	3
3.1. I ² C protokol.....	3
3.2. ATMEGA328P kao master.....	6
3.3. Dizajn kontrolno-upravljačke jedinice.....	6
3.4. Napajanje sklopovlja.....	10
3.5. Napajanje modula	12
4. MODUL – RASVJETNO TIJELO (SLAVE)	13
4.1. ATMEGA168 kao slave.....	13
4.2. Pulsno širinska modulacija (PWM)	14
4.3. Dizajn modula	15
4.4. Princip rada	17
5. INTERAKCIJA, UPRAVLJANJE I KONTROLA.....	19
5.1. Shema master - slave.....	19
5.2. Programsko rješenje.....	20
6. ISPITIVANJE TESTNOG MODELA.....	21
7. ZAKLJUČAK.....	26
POPIS KORIŠTENE LITERATURE	27
POPIS KORIŠTENIH OZNAKA I SIMBOLA	28
SAŽETAK.....	29
ABSTRACT	29
ŽIVOTOPIS	31
PRILOZI.....	32

1. UVOD

Dizajnirati modularno difuzno LED rasvjetno tijelo i pri tom omogućiti i zadržati digitalnu kontrolu i upravljanje intenzitetom svjetla svakog pojedinog modula ili čak svake pojedine LED diode predstavlja izazov u kojem se nameće nekoliko konstrukcijskih i tehničkih problema. Rasvjetno tijelo modularnog dizajna, uz omogućavanje autonomnog digitalnog upravljanja za svaki modul (svako rasvjetno tijelo), a da pri tom zadovoljava tražene uvjete za profesionalnu primjenu, odnosno da moduli pružaju difuznu svjetlost visoke frekvencije i promjenjivog intenziteta, može se koristiti u proizvodnim pogonima kao stabilan izvor svjetlosti za kamere visoke razlučivosti u cilju prepoznavanja anomalija na proizvodima, što bi bila i primarna namjena ovog dizajna. Inženjerski problemi koji se nameću u osmišljavanju i realizaciji ovog projekta kreću od dizajna samog modula (kućište sa elektroničkim sklopovima i konektorima), upravljanja istim, te međusobnoj povezivosti modula i njihovoj komunikaciji sa upravljačkom jedinicom. Da bi ostvarili postojanost i kvalitetu osvjetljenja, ideja je da svaki modul ima vlastito napajanje, a LED diode bile bi pokretane putem sklopa koji na izlazu generira istosmjerni napon visoke frekvencije pomoću PWM modulatora (pulsno širinska modulacija). Uzimajući neupitnu funkcionalnost i stabilnost sklopa kao okvir u odabiru materijala i elektroničkih komponenti istovremeno sam kao bitnu stavku pri izboru istih postavio i cijenu.

2. FUNKCIONALNI BLOK DIJAGRAM



Slika 2.1. Funkcionalni blok dijagram.

Ovaj funkcionalni blok dijagram prikazuje ideju realizacije modularnog LED rasvjetnog tijela sa jednom kontrolno-upravljačkom jedinicom i više upravljivih rasvjetnih modula. Kontrolno-upravljačka jedinica sadržavala bi upravljački sklop sa napajanjem, te integrirano-zasebno napajanje LED dioda za module, dok bi moduli sadržavali upravljački sklop, PWM driver i LED diode (*slika 2.1.*).

3. KONTROLNO-UPRAVLJAČKA JEDINICA (MASTER)

Funkcija kontrolno-upravljačke jedinice je upravljanje LED modulima i kontrola njihovog rada, a kroz projektno rješenje sam u kućište iste integrirao zasebno napajanje LED dioda kako bih pojednostavio izvedbu kompletnog sklopa.

Zamišljeno je da svaki modul ima zasebnu adresu kako bi bio upravljiv nezavisno od ostalih modula u nizu. Upravljačka funkcija mastera je stavljanje modula u rad i regulacija intenziteta svjetla, a kontrolna funkcija je primitak povratnih informacija o stanju modula. Komunikaciju između mastera i modula temeljimo na I²C protokolu.

3.1. I²C protokol

I²C je sinkrona, serijska sabirnica kojoj je namijena međusobno povezivanje raznih digitalnih i upravljivih analognih sustava.

Primjeri digitalnih sustava koji mogu sadržavati I²C sučelje su:

- mikrokontroleri,
- LCD zasloni (engl. *Liquid Crystal Display*),
- EEPROM memorije,
- RAM memorije, itd.

Analogni sustavi koji mogu sadržavati I²C sučelje su:

- pojačala s programibilnim pojačanjem,
- sintetizatore frekvencije, itd.


Komponente se na I²C sabirnicu spajaju izravno, nema potrebe za dodatnim sklopovljem (engl. *glue logic*). Broj komponenata (elemenata) koje se mogu priključiti na sabirnicu, ograničen je jedino najvećim dozvoljenim kapacitetom na svakoj liniji, a on iznosi 400 pF.

Komunikacija putem I²C sabirnica obavlja se putem dvije linije:


- **SDA** (*Serial Data*), kojom se prenose podaci, i
- **SCL** (*Serial Clock*), kojom se prenosi takt.

Naponske razine na ovim linijama gledaju se prema zajedničkoj masi, dok su izlazni stupnjevi komponenata koje se spajaju na sabirnicu u spoju otvorenog kolektora. Iz tog se razloga linije SDA i SCL spajaju na napajanje (+5 V) preko priteznog otpornika (engl. *pull up resistor*).

Koliki će biti iznos otpora priteznog otpornika ovisi o karakteristikama komponenata spojenih na sabirnicu, a u praksi uglavnom iznosi nekoliko kiloohma. Kad je sabirnica slobodna, linije su u stanju logičke jedinice. Napon napajanja na sabirnici je tipično 5V u konkretno ovom slučaju, a može biti i 3,3V, iako standard dozvoljava i više i niže napone. Slike 3.1 i 3.2 prikazuju primjer spajanja komponenata na I²C sabirnicu.



Slika 3.1. Primjer spajanja komponenata na I²C sabirnicu (logička shema).[1]



Slika 3.2. Primjer spajanja komponenata na I²C sabirnicu (električna shema) [2]

Na I²C sabirnici, svaka spojena komponenta mora imati svoju jedinstvenu adresu, a standard definira 7-bitne adrese i time je omogućeno spajanje 128 uređaja sa svojim jedinstvenim adresama. U praksi je moguće spojiti 112 uređaja, dok je preostalih 16 adresa rezervirano za posebne potrebe. Međutim glavni faktor koji određuje maksimalan broj komponenti je definitivno kapacitet na liniji koji sumiran ne smije biti veći od 400pF. Kako dodavanje svake komponente unosi dodatnu kapacitivnost na liniji, ovaj uvjet ograničava najveći broj komponenti na jednoj sabirnici. Svaka komponenta kod I²C može obavljati svoju funkciju na dva načina:

- kao upravljač (engl. *master*), ili
- kao izvršitelj (engl. *slave*).

Upravljač je dominantna komponenta koja kontrolira SCL liniju, odnosno određuje s kojim izvršiteljem će postaviti komunikaciju i koliko će mu dati vremena (određuje takt), uvijek započinje komunikaciju. Izvršitelj (slave) ne može kontrolirati SCL liniju i nikada ne započinje komunikaciju.


Kod I²C protokola, komunikacija se istovremeno ne može odvijati u oba smjera, niti upravljač (master) ne može istovremeno komunicirati sa više od jednog izvršitelja (slave). Upravo zato služi SCL linija koja određuje tko ima prioritet i tko će dobiti vrijeme za komunikaciju. Načini rada sklopova (komponenata) kod I²C protokola su sljedeći:

- slanje podataka od strane upravljača – master upravlja s taktom i šalje podatke izvršitelju,
- upravljač koji prima podatke od izvršitelja – master upravlja s taktom i prima podatke iz izvršitelja,
- slanje podataka od strane izvršitelja – izvršitelj ne upravlja s taktom i šalje podatke upravljaču, ili
- izvršitelj koji prima podatke od upravljača – izvršitelj ne upravlja s taktom i prima podatke od upravljača.

Kako je već navedeno u gornjem tekstu u svakom trenutku komuniciraju jedan upravljač i jedan izvršitelj. Kako bi komunikacija između svih komponenata bila ostvarena, može se dogoditi da neki izvršitelj prijeđe u upravljača i obratno.

3.2. ATMEGA328P kao master

ATMEGA328P je 8 bitni mikrokontroler visokih performansi koji više nego zadovoljava uvjete postavljene u projektnom zadatku. Ideja je da se navedeni mikrokontroler koristi u upravljačko-kontrolnoj jedinici. Komunikacija bi se odvijala putem I²C protokola koji navedeni mikrokontroler podržava na svojim PIN-ovima 27 i 28, što je prikazano na slici 3.3. [3]




Slika 3.3. Raspored nožica na mikrokontroleru ATMEGA328P

3.3. Dizajn kontrolno-upravljačke jedinice


Mozak upravljačke jedinice je Arduino Uno koji zadovoljava sve uvjete postavljene u radnom zadatku i idealan je sklop za upravljanje LED modulima. Arduino Uno je elektronička prototipna platforma namjenjena kreiranju elektroničkih projekata. Sastoji se od *hardware* dijela koji je zapravo fizički elektronički programibilni strujni krug (poznat kao i mikrokontroler) i *software* dijela koji se naziva IDE (*Integrated Development Environment*) kojeg pokrećete na svom računalu i iz njega programirate i upravljate samom pločicom. Sama pločica (slika 3.4.[5]) je nastala u Italiji 2005. godine, a kreirala ga je grupica studenata. Svoju popularnost je stekao prvenstveno zbog jednostavnosti. Samo programiranje pločica ne

zahtjeva dodatan dio hardwarea (koji se zove programer) kao za ostale mikrokontrolerske sustave, nego je dovoljan USB kabel koji se može povezati sa svakim računalnom, bez obzira na operacijski sustav. Arduino možete programirati iz Windowsa, Mac-a, Linuxa, Androida i dr. Postoje i druga rješenja, ali ta platforma uvelike pojednostavljuje izvedbu upravljačke jedinice, te je putem nje vrlo jednostavno isprogramirati mikrokontroler. [4]




Slika 3.4. Arduino Uno


Arduino Uno smješten je u kućište upravljačke jedinice (*slike 3.5. i 3.6.*) dimenzija 240x150x150 mm, zajedno sa napajanjem i korisničkim sučeljem. Upravljačka jedinica ima mogućnost spajanja na računalu putem USB konektora (Comp IN/OUT), te ima jedan izlaz prema modulima (Module OUT) realiziran 4-pinskim konektorom IP68 standarda od kojih 2 pina služe za napajanje i 2 pina za upravljanje modulima, što je vidljivo na slici 3.7. na kojoj je prikazan i A/C priključak 230 V s kućištem osigurača.



Slika 3.5. Prednja strana upravljačke jedinice




Slika 3.6. Stražnja strana upravljačke jedinice



Slika 3.7. Konektori upravljačke jedinice


Korisničko sučelje nalazi se na prednjoj strani kućišta upravljačke jedinice i sadrži sklopku za uključivanje, LED indikator, tipke za odabir modula kojim se želi upravljati, potencijometar za regulaciju intenziteta svjetlosti i LCD zaslon za prikaz relevantnih podataka o odabranom modulu i intenzitetu svjetla u postocima (*slika 3.8.*).



Slika 3.8. Korisničko sučelje

Na shemi spoja (*slika 3.9.*) upravljačke jedinice vidljivo je da se korisničko sučelje u obliku tipki i LCD zaslona spajaju na pinove 0 do 13 arduino platforme što su ustvari digitalni ulazi/izlazi. Napajaju se sa 5 V. Potencijometar za regulaciju intenziteta svjetlosti spaja se na

analogni ulaz A2 jednom nožicom te na napajanje 5 V sa ostale dvije nožice. Putem USB-a se omogućuje ostvarivanje veze sa računalom ukoliko je potrebno, a arduino platforma se napaja sa 12 V iz integriranog napajanja u kućištu koje također napaja i rasvjetne module. Analogni pinovi A4 i A5 služe za komunikaciju sa modulima putem I²C protokola. Napajanje je objašnjeno u sljedećem poglavlju.




Slika 3.9. Shema spoja upravljačke jedinice


3.4. Napajanje sklopovlja

Napajanje je sklop zadužen za opskrbu ostalog sklopovlja dostatnom električnom energijom. U konkretnom slučaju, za napajanje upravljačko-kontrolnih sklopova unutar mastera te napajanje sklopovlja u svakom pojedinom modulu (slaveu), potrebno je osigurati istosmjerni

izvor napajanja od 5V. Istosmjerni napon od 5V za potrebe ovog napajanja izvest ću pomoću integriranog kruga (engl. *IC-Integrated circuit*) LM7805 kao regulatora napona koji će na izlazu isporučiti potrebnih 5V i zadovoljavajućih 1.5A, kako je prikazano na slikama 3.10. i 3.11.



Slika 3.10. Izvedba kućišta i raspored nožica na IC LM7805 [6]



Slika 3.11. Izvedba istosmjernog napajanja napona 5V sa LM7805 [6]

Ovo napajanje, kako piše u naslovu, predviđeno je za potrebe rada upravljačkog dijela mastera i modula, dok se za rad LED dioda u modulima koristi zasebno napajanje.

3.5. Napajanje modula

Iz daljnjih proračuna proizlazi da je potrošnja pojedinog modula približno 22 W. U upravljačku jedinicu integrirano je napajanje snage 360 W, napona 12 V (*slika 3.12 [7].*) iz razloga što se LED diode u ovom slučaju napajaju sa 12 V. Upravljačko sklopovlje modula napaja se sa naponom 5 V koje osigurava gore navedeni integrirani krug LM7805. Na taj način se upravljačko sklopovlje i LED diode napajaju putem istih vodova. Ovim integriranim napajanjem osigurana je stabilna snaga za rad deset modula što je u praksi zadovoljavajuće. U



teoriji je na upravljačku jedinicu moguće spojiti 128 modula ukupne snage približno 2,5 kW, te bi se u tom slučaju moralo osigurati dodatno napajanje.


Slika 3.12. Industrijsko napajanje 360 W, 12 V

4. MODUL – RASVJETNO TIJELO (SLAVE)

Modul je sklop kojem je primarna funkcija isporučiti difuznu svjetlost pomoću svjetlećih LED dioda, s mogućnošću upravljanja intenzitetom svjetla svakog pojedinog modula. Svaki modul ima zasebnu adresu koja bi se određivala od strane korisnika „hardverski“, upotrebom 7 mikroprekidača stanja 0 ili 1, što implicira maksimalni teoretski broj priključenih modula 2^7 odnosno 128. Svojom modularnošću omogućuje fizičku povezanost s kontrolno-upravljačkom jedinicom (masterom) i/ili drugim identičnim modulom (slaveom) u nizu, zadržavajući svoju neovisnost.

4.1. ATMEGA168 kao slave


ATMEGA168 je 8 bitni mikrokontroler visokih performansi koji sa svojih 6 PWM izlaza odgovara cilju, odnosno upravljanjem LED diodama, a cjenovno je prihvatljiviji od mikrokontrolera korištenog u kontrolno-upravljačkoj jedinici. Ideja je da se navedeni mikrokontroler koristi u modulu-rasvjetnom tijelu kao slave. Komunikacija sa kontrolno-upravljačkom jedinicom bi se odvijala putem I²C protokola koji navedeni mikrokontroler podržava na svojim PIN-ovima 27 i 28, što je prikazano na slici 4.1. [4]



Slika 4.1. Raspored nožica na mikrokontroleru ATMEGA168

4.2. Pulsno širinska modulacija (PWM)

Za Pulsno širinsku modulaciju (engl. *Pulse Width Modulation-PWM*), može se reći da je tehnika dobivanja analognih vrijednosti preko digitalnih oblika. Da bi se od digitalnog oblika dobila analogna veličina tj. simulacija analogne veličine, veličinu kojom se želi upravljati treba prikazati kao niz pravokutnih (digitalnih) signala koji se mijenjaju između logičke jedinice (5 V) i logičke nule (0 V). Pošto je digitalni signal diskretan i ima samo dvije razine, amplitudom nije moguće manipulirati, ali se zato može manipulirati sa vremenom u kojem je pravokutni signal u logičkoj jedinici ili nuli. Trajanje vremena u kojem je signal upaljen (logička jedinica), naziva se širina pulsa. Dobivanje traženih analognih vrijednosti postiže se modulacijom širine pulsa. Ako recimo želimo dobiti napon 2.5V, a napon koji je priključen iznosi 5V, širina pulsa treba iznositi 0.5 ili 50% što znači da u jednom taktu logička jedinica treba biti 50% vremena trajanja pulsa. Ako se takav uzorak ponavlja na visokoj frekvenciji kroz neku komponentu (LED dioda), rezultat će biti isti, kao da je dioda spojena na stalni istosmjerni izvor 2.5 V. Slika 4.2 predstavlja primjer PWM-a sa radnim omjerima. [4]




Slika 4.2. Pulsno širinska modulacija (PWM)


Kada se pozove naredba `analogWrite()` generira se pravokutni signal sa određenim radnim omjerom 0 - 255 (256 razina gdje je 255 maksimum iznosa, a 0 minimum) sve do sljedećeg poziva `analogWrite()` ili `digitalRead()` ili `digitalWrite()` na isti pin). Na slici 4.2 prikazane su razine u odnosu sa postocima radnog omjera pa je tako recimo `analogWrite(64)`, radni omjer 25% tj. Ako je maksimalan iznos 5 V, tada bi na izlazu bilo 1.25 V.

4.3. Dizajn modula

U kućište (slike 4.3. i 4.4.) dimenzija 150x100x50 mm smješteno je upravljačko sklopovlje i izvor svjetlosti koji je u ovom slučaju LED dioda. Na gornjoj strani kućišta nalazi se sklopka za uključivanje sa LED indikatorom, 4-pinski konektor IP68 standarda od kojih 2 pina služe za napajanje i 2 pina za komunikaciju sa upravljačkom jedinicom (INPUT), 4-pinski konektor IP68 standarda kao prenosnica napona i signala za sljedeći modul u nizu (OUTPUT) čime osiguravamo modularnost sustava, te adresne mikroprekidače za određivanje jedinstvene fizičke adrese modula u sustavu.



Slika 4.3. Prikaz konektora na LED modulu



Slika 4.4. Prikaz izvora svjetlosti na LED modulu

Za izvedbu izvora svjetlosti koristi se traka sa 90 LED dioda oznake 5050, od kojih svaka daje 50 Lm/W osvjetljenja, sa kutom od 120 stupnjeva, radnim naponom od 12 V i snagom od 240 mW što implicira ukupnu potrošnju od 22 W po modulu pri rad punim intenzitetom. Prednosti pri odabiru ove LED diode ističu se u vrlo otpornoj integriranoj izvedbi u obliku trake, jednostavna montaža i zamjena, cjenovno prihvatljiva i lako dostupna, te apsolutno zadovoljava uvjete osvjetljenja (slika 4.5. [8]). Predviđeni radni vijek je 20.000 sati.




Slika 4.5. Traka LED dioda 5050

Kao difuzor rasvjete korišten je pleksiglas debljine 5 mm. Pleksiglas je razvijen 1928. godine u nekoliko različitih laboratorija. Stručan mu je naziv PMMA što je skraćenica za polimetilmetakrilat. Zbog odličnih tehničkih i kemijskih karakteristika, predstavlja izazov za arhitekta i dizajnere. Iako tehnički nije tip stakla, pleksiglasu se pridodaje naziv akrilno staklo. Kemijski gledano, pleksiglas je sintetički polimer metil-metakrilata. Prozirniji je od običnog stakla te ima odlična optička svojstva jer propušta čak 92% vidljive svjetlosti. Specifična težina mu je 1,17 - 1,20 g/cm³. Pleksiglas je vrlo otporan na vremenske utjecaje i obrada je vrlo jednostavna te se stoga našao na listi za realizacije mnogih ideja na širokom području primjene. [9]

4.4. Princip rada


Sa 4-pinskog konektora INPUT dovodi se upravljački signal sa upravljačke jedinice I²C protokolom na nožice 27 i 28 mikrokontrolera, te napajanje za cijeli modul. Neovisno o modulu, upravljački signali i napajanje prosljeđuju se na 4-pinski konektor OUTPUT za eventualno daljnje spajanje sljedećeg modula u nizu. Za napajanje upravljačkog sklopovlja unutar samog modula koristi se regulator napona LM7805 koji isporučuje napon od dostatnih 5 V. Modul sadržava sklopku za uključivanje u rad sa LED indikatorom, te osigurač od 3 A. Iz sheme (*slika 4.6.*) je vidljivo da se LED diode spajaju na PWM izlaz mikrokontrolera, na nožicu 15, ali preko tranzistora IRFz44n koji na sebe preuzima teret pokretanja LED dioda iz razloga što mikrokontroler inače ne bi mogao podnijeti toliki teret trošila. Tranzistor IRFz44n može isporučiti snagu od 50 W što je dvostruko više od maksimalne potrošnje LED dioda. Tranzistor se hladi pomoću pasivnog aluminijskog hladnjaka. Intenzitet svjetlosti regulira se programskim kodom iz upravljačke jedinice kontroliranjem radnog omjera pulsno širinske modulacije. Jedinstvena fizička adresa se određuje putem mikroprekidača spojenih na nožice 12, 13, 14, 16, 17, 18, 19, a što su digitalni ulazi/izlazi mikrokontrolera.




Slika 4.6. Električna shema modula

5. INTERAKCIJA, UPRAVLJANJE I KONTROLA

5.1. Shema master - slave



Slika 5.1. Shema I²C povezanosti mikrokontrolera ATmega328P (master) i ATmega168 (slave) bez dodatnog napajanja LED dioda [10]



Slika 5.2. Modifikacija sheme modula sa tranzistorima snage i vlastitim napajanjem LED dioda u svrhu veće snage modula


ATMega168 ne može isporučiti dovoljnu snagu da bi LED modul zadovoljavao postavljene uvjete, stoga sam uveo dodatno napajanje i tranzistore snage koji na sebe preuzimaju teret pokretanja većeg broja LED dioda. Napajanje LED dioda osmišljeno je i unaprijed integrirano u kućište kontrolno-upravljačke jedinice.

5.2. Programsko rješenje

Upravljačka jedinica i rasvjetni modul isprogramirani su da funkcioniraju na način kako je opisano u prethodnim poglavljima. Kada se uključi, upravljačka jedinica automatski kreće sa skeniranjem I²C sabirnice i u memoriju sprema adrese pronađenih modula. Korisnik ima opciju odabira modula kojim želi upravljati ili sa svima odjednom. Intenzitet svjetla mijenja se putem potenciometra. Na zaslону upravljačke jedinice moguće je vidjeti koliko modula ima spojenih, njihove adrese, kojim modulom se upravlja te intenzitet osvjetljenja odabranog modula u postocima. Na rasvjetnom modulu, adresa se odabire ručno putem 7 mikroprekidača. Kompletan programski kod moguće je pogledati u prilogima.


6. ISPITIVANJE TESTNOG MODELA

Testni modeli upravljačke jedinice i rasvjetnog modula izvedeni su sa proto pločicama i vizualno ne izgledaju identično kako su zamišljeni u radu. Međutim, u potpunosti su funkcionalni i pomoću njih moguće je dokazati i pokazati da ovaj rasvjetni sustav funkcionira kako je i zamišljeno. Na slikama 6.0 i 6.1 vidljivi su pojedini dijelovi sustava te je objašnjeno što rade.



Slika 6.0 Testni model

Na slici (Slika 6.0) se može vidjeti „mozak“ upravljačke jedinice Arduino uno, spojen sa ostalim sklopovljem upravljačke jedinice koje čine: potencijometar za regulaciju intenziteta, LCD zaslon te tipkalo za upravljanje. Također se vidi testna pločica rasvjetnog modula sa svom elektronikom koja će biti detaljnije objašnjena u daljnjem tekstu.




Slika 6.1 Pločica rasvjetnog modula

Slika 6.1 prikazuje elektroniku rasvjetnog modula. Za napomenuti je korisniku bitan adresni prekidač u desnom kutu. Njime se izabire jedinstvena adresa svakog modula. Osim toga na slici su vidljivi mikrokontroler ATMEGA168 sa potrebnom elektronikom te tranzistor snage IRFz44n na margini pločice, hlađen pasivnim hladnjakom. Rasvjetno tijelo modula vidljivo je na slici ispod (*Slika 6.2*) i čine ga LED diode.



Slika 6.2 Rasvjetno tijelo




Slika 6.3 Prikaz opcija na zaslonu

Slika 6.3 prikazuje korisniku vidljive podatke na zaslonu. Isječak u gornjem lijevom kutu prikazuje mod za pretraživanje rasvjetnih modula, konkretno se vidi nađen jedan modul na adresi 1. Isječak gore desno prikazuje mod za odabir modula te se vidi zadnji postavljen intenzitet svjetlosti svih modula. Isječak dolje lijevo prikazuje upravljanje odabranim modulom, a isječak dolje desno prikazuje upravljanje svim modulima. Ukoliko se uđe u mod za upravljanje svim modulima, automatski se svi rasvjetni moduli postavljaju na zadani intenzitet, ovdje konkretno 36%.



Slika 6.3 Minimalni i maksimalni intenzitet

Slika 6.3 prikazuje razliku između minimalnog i maksimalnog intenziteta. Očita je razlika između minimalnog i maksimalnog intenziteta.



Slika 6.4 Usporedba fotografija slikanih istim fotoaparatom, ali pod različitim osvjetljenjem

Slika 6.4 prikazuje usporedbu fotografije (isječak lijevo) uslikane pod „običnim“ sobnim osvjetljenjem gdje je rasvjetno tijelo klasična žarulja snage 100 watta te fotografije (isječak desno) uslikane uz osvjetljenje jednog rasvjetnog modula. Razlika je očita i namjerno je za pokusni primjerak izabran mobilni uređaj sa razbijenim staklom zaslona. Na desnom isječku puno se bolje vide pukotine i detaljni su bitno izraženiji. Bitno je napomenuti da je i mogućnost uvećavanja na desnom isječku neusporedivo veća nego na lijevom gdje je fotografija i u normalnoj veličini pomalo mutna zbog lošeg osvjetljenja. Na desnom isječku također je moguće vidjeti, u odrazu stakla dio rasvjetnog modula što je dokaz da je za potrebe osvjetljenja upotrijebljen upravo rasvjetni modul, a ne originalan blic fotoaparata.

7. ZAKLJUČAK

Od ideje do realizacije radnog zadatka pojavili su se mnogi izazovi kako hardverski tako i softverski, međutim konačni produkt u testiranjima dao je zavidne rezultate. Upravljačka jedinica ima mogućnost odabira i upravljanja sa svakim rasvjetnim modulom posebno i sa svim modulima odjednom. Upravlja se putem tipki i potenciometra. Potenciometrom se regulira intenzitet svjetla u 255 razina. Povratnu informaciju korisnik ima na 16x2 LCD zaslonu. Na njemu je moguće vidjeti adresu modula kojim se upravlja, adrese svih priključenih modula (maksimalan teoretski broj 128), mod za odabir modula te intenzitet svjetla rasvjetnog modula u postocima. Maksimalan broj rasvjetnih modula koji se mogu priključiti na upravljačku jedinicu bez modifikacije iste je 10. U slučaju potrebe za većim brojem modula javlja se potreba za jačim napajanjem te pojačivačima upravljačkih signala. Jedan modul sa svojih 90 led dioda troši skromih 22 watta električne energije, a isporučuje svjetlost od oko 2000 lumena. Usporedbe radi, klasična žarulja snage 100 watta isporučuje svjetlost od 1600 do 1800 lumena, a troši gotovo četiri puta više električne energije.

POPIS KORIŠTENE LITERATURE

Internet:

- [1] I²C sabirnica, url: http://www.8051projects.net/wiki/I2C_TWI_Tutorial , 8.11.2016.
- [2] I²C sabirnica primjer spajanja, url: <http://maxembedded.com/2014/02/inter-integrated-circuits-i2c-basics/> , 8.11.2016.
- [3] Atmega 328P, url: <http://www.learningaboutelectronics.com/Articles/Atmega328-pinout.php>, 8.11.2016.
- [4] Arduino Uno platforma, url: <https://www.arduino.cc/en/Main/ArduinoBoardUno> , 8.11.2016.
- [5] Arduino Uno, raspored pinova, url: <http://www.electroschematics.com/7958/arduino-uno-pinout/> , 8.11.2016.
- [6] LM7805, url: <https://www.robomart.com/l7805-voltage-regulator> , 12.11.2016.
- [7] Napajanje 360W, url: <http://www.ebay.com/itm/12V-DC-30A-360W-Power-Supply-RepRap-3D-Printer-LED-Strip-Light-CNC-Roboics-/221921250264> , 12.11.2016
- [8] LED traka, url: <http://www.mo-teh.hr/79/led-diskont/led-trake-i-moduli/5050/led-traka-5050-60led-rgb-vodonepropusna/> , 13.11.2016.
- [9] Pleksiglas, url: <http://laserplexx.hr/plexiglas/> , 9.11.2016
- [10] Interakcija (master-slave), url: <http://www.ermicro.com/blog/?p=1239> , 13.11.2016.
- [11] Datasheet Atmega 328P/168, url:
<http://www.alldatasheet.com/view.jsp?Searchword=Atmega328&gclid=CNyM3pbnztACFUVmGwodQPMG5Q>, 29.11.2016.
- [12] Datasheet LM7805, url: <http://www.alldatasheet.com/view.jsp?Searchword=LM7805> , 29.11.2016.
- [13] Datasheet IRFz44n, url:
<http://www.alldatasheet.com/view.jsp?Searchword=IRF%20Z44N> , 29.11.2016.

POPIS KORIŠTENIH OZNAKA I SIMBOLA

Oznaka ili simbol	Naziv	Mjerna jedinica
U	Napon	Volt (V)
I	Struja	Amper (A)
P	Električna snaga	Watt (W)
R	Električni otpor	Ohm (Ω)
T	Vrijeme	Sekunda (S)
l	Dužina	Metar (m)
F	Frekvencija	Hertz (Hz)
C	Električni kapacitet	Farad (F)
Φ_s	Svjetlosni tok	Lumen (Lm)

SAŽETAK

Cilj ovog rada bio je projektirati modularno LED difuzno rasvjetno tijelo s mogućnošću upravljanja intenzitetom svjetlosti svakog pojedinog modula. Cilj je ostvaren na način da je projektirana upravljačka jedinica s korisničkim sučeljem koja upravlja rasvjetnim modulima. Korisničko sučelje upravljačke jedinice, čiji je mozak Arduino uno, omogućava unos osnovnih parametara za upravljanje rasvjetnim modulima, te prikaz istih na LCD zaslonu. Tako odabrane parametre upravljačka jedinica putem I²C protokola isporučuje na odabrani rasvjetni modul, odnosno njegov mikrokontroler ATMEGA168 i regulira intenzitet svjetla. Ovakav sustav osvjetljenja projektiran je prvenstveno za pogone u kojima se zahtjeva osvjetljenje visoke kvalitete. Primjer je kamera visoke razlučivosti koja u nekom pogonu snima proizvod na pogonskoj traci i prepoznaje i najsitnije anomalije. Da bi to bilo izvedivo, osvjetljenje za kameru mora biti savršeno podešeno te ne smije imati negativnog utjecaja na rad kamere. Stoga su ovi rasvjetni moduli opremljeni LED diodama koje su upogonjene pulsno širinskom modulacijom frekvencije od 40kHz jer pri toj frekvenciji kamera ne može raspoznati titranje rasvjetnih tijela koje je pri nižim frekvencijama itekako primjetno kameri. Testiranja koja su obavljena pokazala su da rasvjetni moduli u potpunosti zadovoljavaju uvjete postavljene u radnom zadatku.

Ključne riječi: rasvjetni modul, upravljačka jedinica, LED dioda, Arduino uno, osvjetljenje, mikrokontroler, korisničko sučelje

ABSTRACT

The main goal of this final thesis was to project and design modular LED diffused-light source with ability of tuning intensity of light for each module separately. Objective is accomplished by designing a master control unit with friendly user interface to control light modules. Master unit's user interface, in which is implemented Arduino uno board with ATMEGA328P microcontroller as a „brain“, allows input of basic parameters about module (slave) address and light intensity percentage showing them on LCD screen. Chosen light intensity is afterwards delivered to selected module's microcontroller ATMEGA168, through I²C protocol which further corrects light intensity on LEDs. Such a lightning system is designed for industrial environments and processes which demand high quality light. For instance, a high definition camera which records process on production line with primary task to recognize least anomalies. To avoid and minimize errors, lighting for the camera must be

perfectly set and must not have a negative impact on the operation of the camera. Therefore, all the lighting modules are equipped with LEDs that are driven by PWM frequency of 40 kHz, because at this frequency camera can't recognize flickering lighting that is otherwise visible at lower frequencies. Tests performed, have shown that lightning modules fully meet the conditions set in this task.

Keywords: light modul, control unit, LED diode, Arduino uno, lighting, microcontroller, user interface


ŽIVOTOPIS

Antonio Antunović rođen je 19.06.1993. godine u Našicama gdje je pohađao i završio osnovu školu, srednju školu te školu stranih jezika. Završetkom srednje škole stječe zvanje tehničara za elektroniku. Oduvijek pokazivao zanimanje za elektroniku i mehaniku. Kroz školovanje, uspješno sudjelovao na mnogim projektima vezanima uz elektrotehniku. Nakon srednje škole, upisuje se na Elektrotehnički fakultet u Osijeku gdje namjerava završiti stručni studij automatike. Napredno se služi računalom te poznaje dosta alata vezanih uz elektrotehniku, obradu teksta, slika, videa i glazbe. Ovo su neki od njih: MS Office, Altium designer, Eagle, SketchUP, Multisim, Mathlab, Simulink, Protel, Adobe Photoshop, Sony Vegas te Adobe Audition. Također dobro vlada engleskim jezikom u govoru i pismu. Nakon završetka studija planira tražiti posao u struci.

U Našicama, 29.11.2016.

Antonio Antunović

Potpis:

A rectangular box containing a handwritten signature in blue ink. The signature reads "Antunović A." in a cursive script.

PRILOZI

PRILOG P.1

Vlastiti materijali

PRILOG P.2

Datasheet Atmega328/168 [11]

PRILOG P.3

Datasheet LM7805 [12]

PRILOG P.4

Datasheet IRFz44n [13]

PRILOG P.5

Programski kod za upravljačku jedinicu i rasvjetni modul

Programski kod:**Upravljačka jedinica (ATMEGA328P)**

```

/*
DIFUZNO LED MODULARNO RASVJETNO TIJELO
Name: DR_master_MS (Arduino uno ATMEGA328P) source code
Description: Diplomski_Rad_Master_Multiple_Slaves arduino programski kod
Autor: Antonio Antunović (c)
Date: 24.11.2016.
*/
#include <LiquidCrystal.h>
#include <Wire.h>

int I2CSlaveAddress = 0; // adresa modula za pojedinačno upravljanje
int knobvalue; // varijabla za vrijednost sa potencijometra
int potpin = 2; // pin sa potencijetrom
float percentPWM = 0; // postotak knobvaluea za PWM
int registeredDevices = 0; // broj pronađenih modula u sustavu
int i2cArray[128]; // kapacitivno moguće 127, ali po isporučenoj snazi desetak
/*
// keyboard komunikacija putem tri tipke
// Control - omogućava ulazak u i izazak iz pojedinačne manipulacije modulima
// Scroll - omogućava jednosmjerno listanje kroz adrese pronađenih modula
// Select - omogućava odabir trenutnog modula na listi
*/
const int keyControlPin = 6;
const int keyScrollPin = 7;
const int keySelectPin = 8;
int controlState = 0;
int scrollState = 0;
int selectState = 0;
int ledState = 0;
int tempAddress = 0;
int iCounter = 0;
int selectedState = 0;
int brojacklikova = 0;

const int ledTestPin = 9; // indikator ulaska u Pojedinačno Manipuliranje

/*
LiquidCrystal lcd(RS-pin, E-pin, D4, D5, D6, D7)
- kreiranje liquidcrystal objekta sa pinovima na koje je spojen
- standardni pinovi na Arduino Uno
*/
LiquidCrystal lcd(12, 11, 5, 4, 3, 2);
void setup() {
/*
*****
** Inicijalizacija zaslona **
*****
*/
  lcd.begin(16, 2); // broj kolona, broj redova
  lcd.clear(); // želimo čist ekran pri startanju
  delay(1000);
  lcd.print(" Designed by ...");
  lcd.setCursor(0, 1);
  lcd.print(" Croatia ");
  delay(5000);
  lcd.clear();
  lcd.print("Initializing...");

/*
*****
** Inicijalizacija buttona i PM led-indikatora **
*/

```

```

*****
*/

pinMode(ledTestPin, OUTPUT);
pinMode(keyControlPin, INPUT); // probao i INPUT_PULLUP, nebitno
pinMode(keyScrollPin, INPUT);
pinMode(keySelectPin, INPUT);

/*
*****
** Inicijalizacija I2C protokola **
*****
*/

Wire.begin();
delay(2000);
lcd.clear();

// ovdje ubaciti i2c_scannn i pohraniti pronađene adrese modula u array za daljnje korištenje

I2CSlaveAddress = Find_Slaves(); // idemo potražiti spojene module

if (I2CSlaveAddress == 0) // znači nije nađen niti jedan modul
{
    delay(1000);
    lcd.clear();
    lcd.setCursor(0, 0);
    lcd.print("*** N E M A ***");
    lcd.setCursor(0, 1);
    lcd.print("spojenih modula!");
}
else // znači idemo postaviti prikaz za razradu manipulacije

{
    delay(1000);
    lcd.clear();
    lcd.setCursor(0, 0);
    //lcd.print("0123456789012345");
    lcd.print("Intenzitet ");
    lcd.setCursor(0, 1);
    lcd.print("** SVI MODULI **");
    /*
    lcd.setCursor(10, 1);
    lcd.print(" ");
    lcd.setCursor(10, 1);
    lcd.print(I2CSlaveAddress);
    */
}
}

/*
*****
** MAIN LOOP **
*****
*/

void loop()
{

if (I2CSlaveAddress == 0)
{
    delay(10); //nemoj raditi ništa
}
else

{
    // citaj stanje Control buttona
    controlState = digitalRead(keyControlPin); // sve je naopačke, nije pritisnuto=HIGH
    delay(200); // delay potreban zbog bouncinga
    if (controlState == LOW) // znaci dugmic stisnut
    {
        if (ledState == 0)
        {
            digitalWrite(ledTestPin, HIGH); // ako je dugmic bio stisnut, trebamo to znati, pa
            // uključimo ledicu kao indikator manipulacije modulima
            ledState = 1; // i postavimo stanje varijable na 1 ...
        }
    }
}
}

```

```

iCounter = 0;
I2CSlaveAddress = i2cArray[iCounter];
lcd.setCursor(0, 1);
//lcd.print("0123456789012345");
lcd.print("Listaj >>: "); // ... te obavijestimo korisnika da moz listati po
lcd.setCursor(10, 1); // modulima za pojedinačno upravljanje stavljajući
lcd.print(I2CSlaveAddress); // pocetnu pronadjenu adresu za neki modul u nizu
}
else // u suprotnom dugmic je stisnut, a stanje nase varijable je 1,
{ // znaci treba izaci iz
digitalWrite(ledTestPin, LOW); // pojedinačnog upravljanja i isključiti ledicu kao indikator
ledState = 0; // manipulacije zbog toga i postavljamo nasu varijablu na nulu
selectedState = 0;
lcd.setCursor(0, 1);
lcd.print("* SVI MODULI *"); // i obavjestavamo korisnika da ce upravljati intenzitetom svih modula u nizu
}
}

while ( ledState == 1 ) // dok god je nasa varijabla 1, nalazimo se u pojedinačnoj manipulaciji (PM)
{
controlState = digitalRead(keyControlPin); //provjeriti da kursor nije odustao od PM
if (controlState == LOW)
{
ledState = 0;
selectedState = 0;
iCounter = 0;
digitalWrite(ledTestPin, LOW); // gasi lampicu
lcd.setCursor(0, 1);
lcd.print("* SVI MODULI *");
delay(200);
break;
}
scrollState = digitalRead(keyScrollPin); // idemo očitati stanja scroll tipke i select tipke
selectState = digitalRead(keySelectPin);
delay(200);
if (scrollState == LOW) // isti princip kao i kod controlStatea, znaci scroll je stisnut
{
iCounter++; // brojca mjesta u arrayu popisa registriranih modula povecavamo za jedan
brojacklikova++;
selectedState = 0;
if (iCounter == 127) { // ako je dosao do maksimuma, resetiraj na nula
iCounter = 0;
}
tempAddress = i2cArray[iCounter];
if ( tempAddress == 0 ) { // u slucaju da je temp adresa nula, resetiraj i brojca
iCounter = 0;
tempAddress = i2cArray[iCounter];
}
lcd.setCursor(10, 1);
lcd.print(tempAddress); //prikazi adresu

lcd.setCursor(14, 1);
lcd.print(brojacklikova); //prikazi adresu
}

if (selectState == LOW ) // select tipka je stisnuta
{
// if (selectedState == 0)
// {
I2CSlaveAddress = tempAddress; // trenutna adresa je adresa modula kojim zelimo upravljati
selectedState = 1; // postavljamo signal da je modul izabran
//lcd.print("0123456789012345");
lcd.setCursor(0, 1);
lcd.print("Modul na: "); // obavjestavamo korisnika o adresi izabranog modula
lcd.setCursor(10, 1);
lcd.print(I2CSlaveAddress);
}

if (selectedState == 1) // ako je signal 1, idemo usmjeriti stanje potenciometra prema konkretnom modulu
{
knobvalue = analogRead(potp);
knobvalue = map(knobvalue, 0, 1023, 0, 255);
percentPWM = (knobvalue / 2.55);
lcd.setCursor(11, 0); // pomicanje kursora u prvi red 9 kolonu
lcd.print(" %");
lcd.setCursor(11, 0);
}
}

```



```

    lcd.print(int(percentPWM));
    Wire.beginTransmission(I2CSlaveAddress); // fizička adresa slavea
    Wire.write(knobvalue); // šaljem byte sa količinom duty cycle opterećenja za PWM od 0 do 255
    Wire.endTransmission();
  }
}

if (ledState == 0) // ovdje ćemo poslati vrijednost sa potenciometra na sve module
{
  knobvalue = analogRead(potpin);
  knobvalue = map(knobvalue, 0, 1023, 0, 255);
  percentPWM = (knobvalue / 2.55);
  lcd.setCursor(11, 0); // pomicanje kursora u prvi red 9 kolonu
  lcd.print(" %");
  lcd.setCursor(11, 0);
  lcd.print(int(percentPWM));
  iCounter = 0;
  for (iCounter = 0; iCounter < registeredDevices; iCounter++)
  {
    I2CSlaveAddress = i2cArray[iCounter];
    Wire.beginTransmission(I2CSlaveAddress); // fizička adresa slavea
    Wire.write(knobvalue); // šaljem byte sa količinom duty cycle opterećenja za PWM od 0 do 255
    Wire.endTransmission();
  }
}

}

}

```

```

int Find_Slaves() // vratit ćemo zadnju pronađenu adresu i napuniti i2cArray sa svim adresama na busu
{
  byte error, address, slaveAddress;
  int nDevices;

  lcd.clear();
  lcd.print("Scanning i2c...");
  delay(2000);
  lcd.clear();

  lcd.setCursor(0, 0); // prvi red zaslona
  lcd.print("Adresa:");
  lcd.setCursor(0, 1); // drugi red zaslona
  lcd.print("Modul na:");
  lcd.blink();
  delay(50);
  nDevices = 0;
  for (address = 1; address < 127; address++)
  {
    Wire.beginTransmission(address);
    error = Wire.endTransmission();
    lcd.setCursor(9, 0); // pomicanje kursora u prvi red 9 kolonu
    lcd.print(" ");
    lcd.setCursor(9, 0);
    lcd.print(address);

    if (error == 0)
    {
      nDevices++;
      slaveAddress = address;
      i2cArray[nDevices - 1] = slaveAddress;
      // dodajemo adresu u polje pronađenih adresa radi daljnje manipulacije modulima
      lcd.setCursor(10, 1);
      lcd.print(" ");
      lcd.setCursor(10, 1);
      lcd.print(slaveAddress);
    }

    else if (error == 4)
    {
      lcd.setCursor(9, 0);
      lcd.print("Err");
    }
  }
}

```

```

    lcd.setCursor(13, 0);
    lcd.print(address, HEX);

}
delay(30);
}
registeredDevices = nDevices; //broj registriranih modula na busu
if (nDevices == 0)
{
    slaveAddress = 0;
}
/* DEBUG
    lcd.clear();
    lcd.setCursor(0, 0); // prvi red zaslona
    lcd.print(registeredDevices);
    lcd.setCursor(0, 1); // drugi red zaslona
    lcd.print(i2cArray[0]);
*/
    lcd.noBlink();

return slaveAddress;
}

```

Rasvjetni modul (ATMEGA168)

```

/*
DIFUZNO LED MODULARNO RASVJETNO TIJELO
Name: DR_slave (designed board ATMEGA168) source code
Description: Diplomski_Rad_Slave arduino programski kod
Autor: Antonio Antunović
Date: 24.11.2016.
*/
#include <Wire.h>

int aSwitch1; // digitalpin6
int aSwitch2; // digitalpin7
int aSwitch3; // digitalpin8
int aSwitch4; // digitalpin10
int aSwitch5; // digitalpin11
int aSwitch6; // digitalpin12
int aSwitch7; // digitalpin13
int intAddress = 0 ;

void setup()
{
/*
*****
** Inicijalizacija DIP-SWITCHA **
*****
*/
    pinMode(6, INPUT);
    pinMode(7, INPUT);
    pinMode(8, INPUT);
    pinMode(10, INPUT);
    pinMode(11, INPUT);
    pinMode(12, INPUT);
    pinMode(13, INPUT);

    aSwitch1= digitalRead(6);
    aSwitch2= digitalRead(7);
    aSwitch3= digitalRead(8);
    aSwitch4= digitalRead(10);
    aSwitch5= digitalRead(11);
    aSwitch6= digitalRead(12);
    aSwitch7= digitalRead(13);
    intAddress= aSwitch1+aSwitch2*2+aSwitch3*4+aSwitch4*8+aSwitch5*16+aSwitch6*32+aSwitch7*64;

/*
*****

```

```

** Inicijalizacija I2C protokola i LED trake **
*****
*/
Wire.begin(intAddress); // fizička adresa dobivena DIP-switchem
Wire.onReceive(receiveEvent);
pinMode(9, OUTPUT); // nožica 15 digitalpin 9 je PWM preko kojeg se pogoni LED traka

digitalWrite(9, LOW); // postavljamo stanje na pinu jednako nula ---> nema struje
delay(1000);
digitalWrite(9, HIGH); // postavljamo stanje na pinu jednako jedan ---> ima struje
delay(1000);
digitalWrite(9, LOW);
delay(1000);
digitalWrite(9, HIGH);
delay(1000);
digitalWrite(9, LOW);
delay(1000);
digitalWrite(9, HIGH);
delay(2000);
digitalWrite(9, LOW);
}

/*
*****
** MAIN LOOP **
*****
*/

void loop()
{
  // nula bodova
}

/*
*****
** EVENT HANDLING **
*****
*/
void receiveEvent(int howMany)
{
  while (Wire.available())
  { char c = Wire.read();
    if (c == 'H')
    {
      digitalWrite(9, HIGH); //isporučujemo struju na PWM odnosno LED traku
    }
    else if (c == 'L')
    {
      digitalWrite(9, LOW); // gasimo struju na PWM, odnosno LED traci
    }
    else
    {
      analogWrite(9,c); // analogna vrijednost primljena sa potenciometra
    }
  }
}
}

```